Newsgroup: it.comp.www.asp
Titolo: Numero records per pagina

Ciao a tutti,
eccomi di nuovo per chiedere aiuto :-)
Sto cercando di far si che su una pagina vengano
visualizzati solo 10 record alla volta. Questa operazione
sembra essere riuscita, ma c'è un problema che proprio non
riesco a capire. Infatti, nella pagina vengono visualizzati
10 record ma tutti uguali !!!!
Qualcuno potrebbe darmi una mano ad evidenziare l'errore nel
codice?
Se volete vedere la pagina in azione la trovate a questo
indirizzo:
http://207.5.16.60/soluzioni.asp
Ed ora il codice...
Grazie
Omar
<%
    'Set how many records per page we want
    Const NumPerPage = 10
    'Retrieve what page we're currently on
    Dim CurPage
    If Request.QueryString("CurPage") = "" then
        CurPage = 1 'We're on the first page
    Else
        CurPage = Request.QueryString("CurPage")
    End If
    Dim connSearch
    Set connSearch = Server.CreateObject("ADODB.Connection")
    connSearch.Open "soluzioni"
    'Explicitly Create a recordset object
    Dim rsSearch
    Set rsSearch = Server.CreateObject("ADODB.Recordset")
    'Set the cursor location property
    rsSearch.CursorLocation = adUseClient
    'Set the cache size = to the # of records/page
    rsSearch.CacheSize = NumPerPage
 Dim cerca
 cerca = Request.Form("qry")
 if cerca = "" then cerca = "%"
    'Open our recordset
    Dim searchSQL
    searchSQL = "SELECT * FROM tabSoluzioni WHERE Titolo
LIKE '%"&cerca&"%'"
 searchSQL = searchSQL & "ORDER BY Titolo ASC;"
 rsSearch.Open searchSQL, ConnSearch
    rsSearch.MoveFirst
    rsSearch.PageSize = NumPerPage
    'Get the max number of pages
    Dim TotalPages
    TotalPages = rsSearch.PageCount
    'Set the absolute page
    rsSearch.AbsolutePage = CurPage
%>
<table cellpadding="0" cellspacing="2" border="0">
<tr>
 <td colspan="2">
 <%
 'Print out the current page # / total pages
    Response.Write FONT1
 Response.Write("Pagina " & CurPage & " di " & TotalPages)
 %>
 </td>
</tr>
<tr>
 <td width="400" align="center" bgcolor="#C5C5E2">
 <% Response.Write FONT1 %><b>Titolo</b></font></td>
 <td width="150" align="center" bgcolor="#C5C5E2">
 <% Response.Write FONT1 %><b>Lingua</b></font></td>
</tr>
<%
    'Counting variable for our recordset
    Dim count
    'Set Count equal to zero
    Count = 0
 Dim titolo
 Dim lingua
 Dim cat
 Dim id
 titolo = rsSearch("Titolo")
 lingua = rsSearch("Lingua")
 cat = rsSearch("Categoria")
 id1 = rsSearch("id")
    Do While Not rsSearch.EOF And Count < rsSearch.PageSize
        Count = Count + 1
  Response.Write "<tr><td><b>"
  Response.Write FONT1
  Response.Write "<a
href=testo_soluz.asp?cat="&cat&"&ID="&id1&">"
  Response.Write titolo
  Response.Write "</a></font></b></td>"
  Response.Write "<td>"
  Response.Write FONT1
  Response.Write lingua
  Response.Write "</font></td></tr>"
        rsSearch.MoveNext
    Loop
 Response.Write "</table>"
    'Display Next / Prev buttons
    if CurPage > 1 then
        'We are not at the beginning, show the prev button
        Response.Write("<INPUT TYPE=BUTTON VALUE=Precedente
ONCLICK=""document.location.href='search.asp?curpage=" &
curpage - 1 & "';"">")
    End If
 Response.Write "&nbsp;&nbsp;&nbsp;"
 if CInt(CurPage) <> CInt(TotalPages) then
        'We are not at the end, show a next button
        Response.Write("<INPUT TYPE=BUTTON VALUE=Prossimo
ONCLICK=""document.location.href='search.asp?curpage=" &
curpage + 1 & "';"">")
    End If
rsSearch.Close
Set rsSearch = Nothing
ConnSearch.Close
Set ConnSearch = Nothing
%>

    Loop
Metti la riga con l'inizio ciclo ( Do While Not rsSe...) PRIMA di  titolo =
rsSearch("Titolo"), in questo modo:
Do While Not rsSearch.EOF And Count < rsSearch.PageSize
Count = Count + 1
titolo = rsSearch("Titolo")
lingua = rsSearch("Lingua")
cat = rsSearch("Categoria")
id1 = rsSearch("id")
' e tutto il resto...
  Response.Write "<tr><td><b>"
  Response.Write FONT1
  Response.Write "<a
href=testo_soluz.asp?cat="&cat&"&ID="&id1&">"
  Response.Write titolo
  Response.Write "</a></font></b></td>"
  Response.Write "<td>"
  Response.Write FONT1
  Response.Write lingua
  Response.Write "</font></td></tr>"
        rsSearch.MoveNext
    Loop
Non avrai più problemi!
Ciao!
Andrew